


Linking Latin Philosophical Expressions

Giovanna Tonazzo



Agenda

Motivation

Methodology

Results

Data and code are available on [GitHub](#)



Motivation



Latin philosophical expressions

Latin expressions like *"cogito ergo sum"* or *"homo homini lupus"* are found in nearly every philosophy manual. These expressions are often:

- Central to a philosopher's worldview
- Full phrases or sentences, not just isolated words
- Rarely covered comprehensively in standard dictionaries

Philosophy students sometimes lack a strong background in Latin, making it difficult to:

- Grasp literal meanings of these expressions
- Understand their grammatical structures
- Connect their linguistic form with philosophical content

This project is inspired by a [resource](#) created by Prof. James Lesher – a curated collection of Latin philosophical expressions. Each expression in the collection includes:

- A literal translation into English
- Linguistic analysis (parts of speech, case, syntax)
- A philosophical explanation, often linked to thinkers like Aristotle, Descartes, or Kant
- Examples and context from historical texts

Examples of Latin expressions from the collection

A posteriori: preposition + the ablative neuter singular of the comparative adjective *posterior/posteriorus* (literally: ‘from the later thing’): things known *a posteriori* are known on the basis of experience (e.g. ‘We can know only *a posteriori* that all swans are white’).

A priori: preposition + the ablative neuter singular of the comparative adjective *prior/prius* (literally: ‘from the earlier thing’): what is known to be true *a priori* can be known independently of (or prior to) empirical investigation or confirmation (e.g. ‘Kant held that we can know *a priori* that a straight line is the shortest distance between two points.’)

Amicus Plato sed magis amica veritas: ‘Plato is a friend but truth is a greater friend’, based loosely on Aristotle, *Nicomachean Ethics* 1096a.

Can this resource be improved?

This resource is very useful for students, but it has some limitations:



It gives a short definition of each expression, but many of them have been interpreted in different ways by different thinkers. Expanding the entries to indicate key philosophical interpretations would help student develop a critical understanding of concepts.



When philosophers are mentioned, it does not provide direct links or references to their works or biographical information. Adding these connections would make it easier for students to place the terms in their proper philosophical context.



Some entries are related (i.e “a priori” and “a posteriori”) but these relationships are not explicitly indicated in the document. Making these connections visible would help students understand the logical and conceptual relationships among terms.

Can this resource be improved?



This resource exists in a relatively **isolated** form. Publishing it as **linked data** by integrating it with existing ontologies or connecting entries to structured resources like Wikidata would significantly enrich its role as a resource for philosophical study.



This would allow students to situate each term within a broader conceptual and historical **network**, and to **explore** philosophical concepts across languages and interpretations more effectively.



Linking terms, expressions, and philosopher names to external sources would transform the resource from a static glossary into a dynamic, **interoperable** tool for philosophical study and research.



Methodology



Steps taken to enrich the resource (1)

- Converted the original textual content into a structured format (CSV).
- Added additional expressions (e.g., *homo homini lupus*); the final version is not fully aligned with Prof. Leshner's original collection.
- Provided more extensive definitions (with the help of ChatGPT – see [notebook](#)) to clarify the general philosophical concepts.
- Classified expressions by branch of philosophy (e.g., epistemology, ethics, logic).
- Included alternative interpretations from different philosophers.

Steps taken to enrich the resource (2)

- Linked expressions, philosophers, and philosophical branches to external resources (Wikidata).
- Created internal links between related expressions (e.g., *a priori* and *a posteriori*, *res cogitans* and *res extensa*).
- Performed grammatical analysis (tokenization, part of speech tagging, morphological features, dependency relations) following the Universal Dependencies framework.
- Linked Latin lemmas to the LiLa “Linking Latin” knowledge base.
- Published as linguistic linked data (triplestore in Turtle format).

Tools used

- OpenRefine for reconciliation with Wikidata page of entry, philosopher, branch of philosophy.
- Custom Python procedures
 - to call LiLa TextLinker (see [notebook](#)).
 - to call UDPipe API for annotation.
 - to generate the triplestore using RDFLib (see [notebook](#)).
- Apache Jena Fuseki and LodLive for visualization.
- Basic [SPARQL queries](#) for analysis.

Linking to LiLa

Problems:

- No matches: solved by manual search in Lemma Bank query interface
- More than one match: ambiguity resolved with a SPARQL query to go from lemma to Wordnet Synset to clarify meaning

lila-erc.eu/sparql/

LILA: SPARQL ACCESS POINT

PREFIXES:

Query × Query 1 × Query 2 × Query 3 × +

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX ontollex: <http://www.w3.org/ns/lemon/ontollex#>
3 PREFIX lila: <http://lila-erc.eu/ontologies/lila/>
4 PREFIX wn: <http://wordnet-rdf.princeton.edu/ontology#>
5 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
6
7 SELECT DISTINCT ?wrp ?pos ?lexical_entry ?synset_definition WHERE {
8   ?lexical_entry ontollex:writtenRep ?wrp . FILTER regex(?wrp, "^fundo$", "i") .
9   ?lexical_entry lila:hasPOS ?pos .
10  ?canonical_form ontollex:canonicalForm ?lexical_entry .
11  ?canonical_form ontollex:evokes ?synset .
12  ?synset rdfs:label ?synset_label ;
13    skos:definition ?synset_definition .
14  FILTER regex(?synset_label, "^Wordnet Synset.*")
15 }
16
```

Table Response 29 results in 0.47 seconds

| wrp | pos | lexical_entry | synset_definition |
|-------|-----------|---|---------------------------------|
| fundo | lila:verb | http://lila-erc.eu/data/id/lemma/104012 | spend thoughtlessly; throw away |
| fundo | lila:verb | http://lila-erc.eu/data/id/lemma/104013 | make more firm |
| fundo | lila:verb | http://lila-erc.eu/data/id/lemma/104013 | set up or found |

Input data for Turtle file

- CSV files for entries and interpretations
- JSON file for LiLa links
- Manually curated list of linked concepts
- Response in CoNLL-U format returned by UDPipe (using model "latin-evalatin24-240520")

Modeling

I considered the resource both as a corpus and as a lexical resource.

Considering it as a corpus, I modeled it using **POWLA**. This allowed a modular approach that made possible to distinguish between content layers and annotation layers. For the syntactic annotation, I used also Web Annotation.

Content layers:

- Each expression is a root that belongs to the Expression Layer
- The tokens are terminals in the Document Layer

Annotation layers:

- Dependency Annotation Layer contains the syntactic dependency relations
- UD Annotation Layer contains POS and morphological features
- Semantic Annotation Layer to connect entry to its general meaning

Modeling

Considering the collection as a lexical resource, I modeled it using **ONTOLEX**.

LIME lexicons

- Separate LIME lexicons for Latin and English
- Each Latin expression is a lexical entry in the Latin lexicon
- The corresponding English translation is an entry in the English lexicon

VARTRANS translation

- The translation is a relation between Latin and English lexical entries
- The relation is reified and it is of category “philosophicalTranslation”


Modeling

Semantics

- General concept modeled as lexical concept evoked by the lexical entry (both Latin and English lexical entry evoke the same concept)
- The lexical concept has a SKOS definition and link to Wikidata page
- Branch of philosophy linked to lexical concept as broader concept
- Relations between expressions (i.e. "a priori" and "a posteriori") as relations between lexical concepts
- Specific interpretations by different philosophers as different senses, linked to the general concept via SKOS narrower relation
- Philosopher are connected to the sense and to the lexical concept as dc:creator


POWLA corpus

- Content Layers
- Annotation Layers




Content Layers

- Expression Layer: expressions as POWLA roots
- Document Layer: tokens as POWLA terminals, prev-next for linear ordering




Syntax

- Dependency Annotation Layer
- Each node is a deprel that connects source (head) to target (dependent)



POS and morphology

- UD Annotation Layer
- Each note has the token as target and the POS/morphological features as body
- POS and features are linked to UD definition page
- Each token has a lemma connected to LiLa lemma



ONTOLEX: LIME and VARTRANS

- Each expression is a LexicalEntry of a lexicon
- Latin lexicon and English lexicon
- Translation as LexicalRel between lexical entries
- Translation is reified and it is of category “philosophicalTranslation”



Semantics

- Lexical Entry evokes a Lexical Concept
- Branch of philosophy as broader concept
- Links to entry, philosopher and branch in Wikidata
- Interpretations by philosophers are lexical senses




Interpretations

- Different interpretations by philosophers as lexical senses
- Narrower concepts of lexical concept
- Philosopher as dc:creator



Related concepts

SKOS:related





Results



Quantitative results

| | # | SPARQL Query |
|---------------------------------------|--------|---|
| Number of expressions in each lexicon | 153 | <pre>SELECT ?lexicon (COUNT(?philoeexprid) AS ?expr_count) WHERE { ?philoeexprid rdf:type ontolox:LexicalEntry; lime:entry ?lexiconid . ?lexiconid rdfs:label ?lexicon. } GROUP BY ?lexicon</pre> |
| Number of triples | 18 780 | <pre>SELECT ?subject ?predicate ?object WHERE { ?subject ?predicate ?object }</pre> |
| Number of philosophers | 109 | <pre>SELECT (COUNT(DISTINCT ?philosopherid) AS ?philosophers) WHERE { ?senseid a ontolox:LexicalSense ; dct:creator ?philosopherid }</pre> |
| Number of senses | 614 | <pre>SELECT (COUNT(?senseid) AS ?senses) WHERE { ?senseid a ontolox:LexicalSense . }</pre> |

Examples of analysis: From LiLa to philosophy

```
7 SELECT DISTINCT ?wrp ?philoepr ?expression_wiki ?philosopher ?philosopher_wiki
8 WHERE {
9   SERVICE <https://lila-erc.eu/sparql/lila_knowledge_base/sparql> {
10     ?lilalemma ontolex:writtenRep ?wrp . FILTER regex(?wrp, "^homo$", "i") .
11   }
12   ?terminal lila:hasLemma ?lilalemma.
13   ?expression powla:hasTerminal ?terminal ;
14     rdfs:label ?philoepr .
15   ?expression ontolex:evokes ?concept.
16   ?concept dct:creator ?creator ;
17     rdfs:seeAlso ?expression_wiki.
18   ?creator rdfs:label ?philosopher ;
19     rdfs:seeAlso ?philosopher_wiki.
20 }
21 ORDER BY ?philoepr
22
23
```

Table Response 20 results in 0.698 seconds

Simple view ☐ Ellipse ☒ Fil

| wrp | philoepr | expression_wiki | philosopher | philosopher_wiki |
|------------|----------------------------|--|-------------------------|---|
| 1"homo"@la | "argumentum ad hominem"@la | <http://www.wikidata.org/entity/Q189183> | "Aristotle"@en | <https://www.wikidata.org/wiki/Q84473023> |
| 2"homo"@la | "argumentum ad hominem"@la | <http://www.wikidata.org/entity/Q189183> | "David Hume"@en | <https://www.wikidata.org/wiki/Q37160> |
| 3"homo"@la | "argumentum ad hominem"@la | <http://www.wikidata.org/entity/Q189183> | "Immanuel Kant"@en | <https://www.wikidata.org/wiki/Q9312> |
| 4"homo"@la | "argumentum ad hominem"@la | <http://www.wikidata.org/entity/Q189183> | "John Stuart Mill"@en | <https://www.wikidata.org/wiki/Q50020> |
| 5"homo"@la | "homo faber"@la | <http://www.wikidata.org/entity/Q422282> | "Hannah Arendt"@en | <https://www.wikidata.org/wiki/Q60025> |
| 6"homo"@la | "homo faber"@la | <http://www.wikidata.org/entity/Q422282> | "Karl Marx"@en | <https://www.wikidata.org/wiki/Q9061> |
| 7"homo"@la | "homo faber"@la | <http://www.wikidata.org/entity/Q422282> | "Martin Heidegger"@en | <https://www.wikidata.org/wiki/Q48301> |
| 8"homo"@la | "homo faber"@la | <http://www.wikidata.org/entity/Q422282> | "Simone de Beauvoir"@en | <https://www.wikidata.org/wiki/Q7197> |

Expressions containing "homo"





Interpretations of "Summum bonum"

```
3 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
4 PREFIX skos: <http://www.w3.org/2004/02/skos/core#>
5 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
6 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
7 PREFIX ontolex: <http://www.w3.org/ns/lemon/ontolex#>
8
9 SELECT ?philosopher ?definition
10 WHERE {
11   ?form ontolex:writtenRep ?wr.
12   FILTER (?wr = "Summum bonum"@la).
13   ?expression rdf:type ontolex:LexicalEntry ;
14               ontolex:canonicalForm ?form .
15   ?expression ontolex:sense ?sense.
16   ?sense skos:definition ?definition ;
17          dct:creator ?creator .
18   ?creator rdfs:label ?philosopher .
19 }
```

Table Response 4 results in 0.012 seconds

| philosopher | definition |
|-----------------------|--|
| 1 "Immanuel Kant"@en | "Kant interprets 'summum bonum' as the highest good, which combines both virtue and happiness. He posits that moral actions should lead to ha |
| 2 "Nietzsche"@en | "Nietzsche challenges traditional notions of 'summum bonum', arguing that the concept is a construct of slave morality that suppresses individual |
| 3 "Aristotle"@en | "For Aristotle, the 'summum bonum' is the ultimate goal of human life, which he identifies as eudaimonia, often translated as 'happiness' or 'flouri |
| 4 "Utilitarianism"@en | "Utilitarian philosophers like Bentham and Mill interpret 'summum bonum' as the greatest happiness principle. They argue that the moral worth of |

Translation and definition of "Homo homini lupus"

| <pre>3 PREFIX rdfrs: <http://www.w3.org/2000/01/rdf-schema#> 4 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#> 5 PREFIX ontolx: <http://www.w3.org/ns/lemon/ontolx#> 6 7 SELECT ?wr ?wrEn ?def 8 WHERE { 9 ?form ontolx:writtenRep ?wr. 10 FILTER (?wr = "Homo homini lupus"@la). 11 ?expression rdf:type ontolx:LexicalEntry ; 12 ontolx:canonicalForm ?form . 13 ?expression ontolx:evokes ?concept. 14 ?concept skos:definition ?def. 15 ?trans vartrans:source ?expression ; 16 vartrans:target ?expressionEn . 17 ?expressionEn ontolx:canonicalForm ?formEn . 18 ?formEn ontolx:writtenRep ?wrEn. 19 }</pre> | | | 1 result in 0.015 seconds | | Simple view <input type="checkbox"/> Ellipse <input checked="" type="checkbox"/> Filter query results | Page size: 50 |   |
|---|------|--|---------------------------|--|---|---------------|---|
| wr | wrEn | def | | | | | |
| 1"Homo homini ... "Man is a wolf to man"@en | | "The Latin phrase 'homo homini lupus' translates to 'man is a wolf to man.' This concept, attributed to the Roman playwright Plautus and later popularized by the philosopher Thomas Hobbes, suggests that humans, in their natural state, are prone to aggression and selfishness, akin to predatory animals. The phrase encapsulates a cynical view of human nature, emphasizing that individuals are often in competition with one another, leading to conflict and violence. Hobbes, in his work 'Leviathan,' argued that without a strong central authority to impose order, life would be 'solitary, poor, nasty, brutish, and short.' This perspective contrasts with more optimistic views of human nature that emphasize cooperation and altruism. The phrase serves as a reminder of the darker aspects of humanity, suggesting that social structures and moral codes are necessary to mitigate our inherent tendencies toward hostility and self-interest."@en | | | | | |

Philosophers connected to most expressions

```
6 SELECT ?philosopher (COUNT(?philoxpr) AS ?expr_count)
7 WHERE {
8   ?concept rdf:type ontolox:LexicalConcept ;
9             dct:creator ?creator .
10  ?creator rdfs:label ?philosopher .
11  ?expression ontolox:evokes ?concept ;
12             rdfs:label ?philoxpr.
13  FILTER (lang(?philoxpr) = 'la')
14 }
15 GROUP BY ?philosopher
16 ORDER BY DESC(?expr_count)
17
```

Table Response 109 results in 0.02 seconds

Sim

| philosopher | expr_count |
|------------------------|--|
| 1"Immanuel Kant"@en | "88"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 2"Aristotle"@en | "69"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 3"Nietzsche"@en | "61"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 4"David Hume"@en | "57"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 5"Thomas Aquinas"@en | "49"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 6"Leibniz"@en | "45"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 7"Descartes"@en | "17"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 8"John Stuart Mill"@en | "17"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 9"Martin Heidegger"@en | "12"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 10"Plato"@en | "11"^^<http://www.w3.org/2001/XMLSchema#integer> |

Expressions connected to Aristotle

```
9 SELECT ?philosopher ?philoexpr
10 WHERE {
11     ?concept    rdf:type ontolex:LexicalConcept ;
12                dct:creator ?creator .
13     ?creator    rdfs:label ?philosopher FILTER regex(?philosopher, "^Aristotle$", "i") .
14     ?expression ontolex:evokes ?concept ;
15                rdfs:label ?philoexpr.
16     FILTER (lang(?philoexpr) = 'la')
17 }
18 ORDER BY ?philoexpr
19
```

Table Response 69 results in 0.012 seconds

| philosopher | philoexpr |
|------------------|--|
| 1"Aristotle"@en | "a dicto secundum quid ad dictum simpliciter"@la |
| 2"Aristotle"@en | "a dicto simpliciter ad dictum secundum quid"@la |
| 3"Aristotle"@en | "a fortiori"@la |
| 4"Aristotle"@en | "a necesse ad esse valet consequentia"@la |
| 5"Aristotle"@en | "ab esse ad posse valet consequentia"@la |
| 6"Aristotle"@en | "ab ovo"@la |
| 7"Aristotle"@en | "actus purus"@la |
| 8"Aristotle"@en | "ad hoc"@la |
| 9"Aristotle"@en | "ad infinitum"@la |
| 10"Aristotle"@en | "amicus Plato sed magis amica veritas"@la |

Expressions belonging to "Logic"

```
9 SELECT DISTINCT ?branch_label ?branch_wiki ?expression ?definition ?concept_wiki
10 WHERE {
11     ?branch skos:prefLabel ?branch_label FILTER regex(?branch_label, "^Logic$", "i") .
12     ?branch a skos:Concept ;
13         rdfs:seeAlso ?branch_wiki .
14     ?concept skos:broader ?branch ;
15         skos:definition ?definition .
16     ?concept rdfs:seeAlso ?concept_wiki .
17     ?expr ontalex:evokes ?concept ;
18         a powla:root ;
19         rdfs:label ?expression .
20 }
```

Table Response 26 results in 0.022 seconds

Simple view ☐ Ellipse ☒ Filter query results Page size: 50

| branch_label | branch_wiki | expression | definition | concept_wiki |
|--------------|--|------------|--|---|
| 1"Logic"@en | < https://www.wiki... "characteristica universalis"... | | "The term 'characteristica universalis' refers to a philosophical and lin... | < http://www.wikidata.org/entity/Q2142306 > |
| 2"Logic"@en | < https://www.wiki... "nihil obstat"@la | | "Nihil obstat is a Latin phrase meaning 'nothing hinders' or 'nothing s... | < http://www.wikidata.org/entity/Q1994297 > |
| 3"Logic"@en | < https://www.wiki... "a fortiori"@la | | "The Latin phrase 'a fortiori' translates to 'from the stronger' and is us... | < http://www.wikidata.org/entity/Q1753631 > |
| 4"Logic"@en | < https://www.wiki... "salva veritate"@la | | "The Latin phrase 'salva veritate' translates to 'saving the truth' or 'wit... | < http://www.wikidata.org/entity/Q11892155 > |
| 5"Logic"@en | < https://www.wiki... "argumentum ad populum"... | | "Argumentum ad populum, also known as the appeal to the masses ... | < http://www.wikidata.org/entity/Q251695 > |
| 6"Logic"@en | < https://www.wiki... "petitio principii"@la | | "Petitio principii, often translated as 'begging the question', is a logic... | < http://www.wikidata.org/entity/Q219429 > |
| 7"Logic"@en | < https://www.wiki... "dictum de omni et nullo"@la | | "Dictum de omni et nullo is a Latin phrase that translates to 'a statem... | < http://www.wikidata.org/entity/Q12764637 > |
| 8"Logic"@en | < https://www.wiki... "argumentum ad hominem"... | | "Argumentum ad hominem is a Latin phrase that translates to 'argum... | < http://www.wikidata.org/entity/Q189183 > |

Linked concepts

```
8 SELECT ?philoepr1 ?philoepr2
9 WHERE {
10   ?concept1 rdf:type ontolox:LexicalConcept .
11   ?concept2 rdf:type ontolox:LexicalConcept .
12   ?concept1 skos:related ?concept2 .
13   ?expression1 ontolox:evokes ?concept1 ;
14               rdfs:label ?philoepr1.
15   ?expression2 ontolox:evokes ?concept2 ;
16               rdfs:label ?philoepr2.
17   FILTER (lang(?philoepr1) = 'la')
18   FILTER (lang(?philoepr2) = 'la')
19 }
20
```

[Table](#) [Response](#) 44 results in 0.014 seconds

| philoepr1 | philoepr2 |
|----------------------------------|--|
| 1"tabula rasa"@la | "nihil in intellectu nisi prius in sensu"@la |
| 2"mundus sensibilis"@la | "mundus intelligibilis"@la |
| 3"bellum omnium contra omnes"@la | "homo homini lupus"@la |
| 4"a fortiori"@la | "a priori"@la |
| 5"a fortiori"@la | "a posteriori"@la |
| 6"per se"@la | "per accidens"@la |
| 7"per accidens"@la | "per se"@la |
| 8"argumentum ad populum"@la | "argumentum ad hominem"@la |
| 9"argumentum ad populum"@la | "argumentum ad baculum"@la |
| 10"de jure"@la | "de re"@la |

Most frequent POS and Nouns

```
13 SELECT ?pos_id (COUNT(?token_id) AS ?token_count)
14 WHERE {
15   ?pos rdfs:label ?pos_id. FILTER regex(?pos_id, "^POS=", "i") .
16   ?annot oa:hasBody ?pos .
17   ?annot oa:hasTarget ?token_id .
18 }
19 GROUP BY ?pos_id
20 ORDER BY DESC(?token_count)
21
```

Table Response 13 results in 0.018 seconds

| pos_id | token_count |
|-------------|---|
| 1POS=NOUN | "148"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 2POS=ADP | "60"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 3POS=VERB | "58"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 4POS=ADJ | "57"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 5POS=PRON | "29"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 6POS=DET | "19"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 7POS=ADV | "16"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 8POS=AUX | "15"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 9POS=PART | "11"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 10POS=SCONJ | "6"^^<http://www.w3.org/2001/XMLSchema#integer> |

```
5 SELECT ?lemma (COUNT(?lemma) as ?occurrences)
6 WHERE {
7   ?pos rdfs:label ?pos_id. FILTER regex(?pos_id, "^POS=NOUN", "i") .
8   ?annot oa:hasBody ?pos .
9   ?annot oa:hasTarget ?token_id .
10  ?token_id lila:hasLemma ?lemma_id .
11  ?lemma_id rdfs:label ?lemma .
12 }
13 GROUP BY ?lemma
14 ORDER BY DESC(?occurrences)
```

Table Response 95 results in 0.021 seconds

| lemma | occurrences |
|--------------|---|
| 1homo | "6"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 2res | "6"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 3causa | "5"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 4intellectus | "5"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 5dictum | "4"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 6ens | "4"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 7argumentum | "3"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 8modus | "3"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 9mundus | "3"^^<http://www.w3.org/2001/XMLSchema#integer> |
| 10natura | "3"^^<http://www.w3.org/2001/XMLSchema#integer> |

UD annotations

```
8 SELECT ?token ?deprel ?deprel ?headtoken (GROUP_CONCAT(CONCAT(?feat); separator="|") AS ?feats)
9 WHERE {
10   ?form ontolex:writtenRep ?wr.
11   FILTER (?wr ="Entia non sunt multiplicanda praeter necessitatem"@la).
12   ?root rdf:type powla:root ;
13         ontolex:canonicalForm ?form .
14   ?root powla:hasTerminal ?tokenid .
15   ?tokenid rdfs:label ?token .
16   OPTIONAL {
17     ?deprelid powla:hasSource ?headtokenid ;
18             powla:hasTarget ?tokenid ;
19             rdfs:label ?deprel .
20     ?headtokenid rdfs:label ?headtoken
21   } .
22   ?annotid oa:hasTarget ?tokenid ;
23           oa:hasBody ?featid .
24   ?featid rdfs:label ?feat .
25 }
26 GROUP BY ?tokenid ?token ?deprel ?deprel ?headtoken
27 ORDER BY ?tokenid ?feat
28
```

Table Response 6 results in 0.014 seconds

| token | deprel | headtoken | feats |
|---------------------|--------------------|--------------------|---|
| 1"entia"@la | deprel: nsubj:pass | "multiplicanda"@la | Number=Plur POS=NOUN Case=Nom Gender=Neut Inflclass=Indeuri |
| 2"non"@la | deprel: advmod:neg | "multiplicanda"@la | Polarity=Neg POS=PART |
| 3"sunt"@la | deprel: aux:pass | "multiplicanda"@la | Number=Plur Aspect=Imp Tense=Pres Mood=Ind Person=3 Verbform=Fin POS=AUX |
| 4"multiplicanda"@la | | | Number=Plur Verbform=Part Case=Nom Gender=Neut POS=VERB Aspect=Prosp Voice=Pass |
| 5"praeter"@la | deprel: case | "necessitatem"@la | Adptype=Prep POS=ADP |
| 6"necessitatem"@la | deprel: obl | "multiplicanda"@la | Case=Acc Number=Sing POS=NOUN Gender=Fem |

```
7 SELECT ?token ?feat ?udfeatid
8 WHERE {
9   ?form ontolex:writtenRep ?wr.
10  FILTER (?wr ="Homo homini lupus"@la).
11  ?root rdf:type powla:root ;
12        ontolex:canonicalForm ?form ;
13        powla:hasTerminal ?tokenid .
14  ?tokenid rdfs:label ?token .
15  ?annotid oa:hasTarget ?tokenid ;
16          oa:hasBody ?featid .
17  ?featid rdfs:label ?feat ;
18          rdf:type ?udfeatid .
19 }
20 ORDER BY ?tokenid ?feat
21
```

Table Response 12 results in 0.01 seconds

| token | feat | udfeatid |
|--------------|-------------|---|
| 1"homo"@la | Case=Nom | < https://universaldependencies.org/u/feat/Case#Nom > |
| 2"homo"@la | Gender=Masc | < https://universaldependencies.org/u/feat/Gender#Masc > |
| 3"homo"@la | Number=Sing | < https://universaldependencies.org/u/feat/Number#Sing > |
| 4"homo"@la | POS=NOUN | < https://universaldependencies.org/u/pos/NOUN > |
| 5"homini"@la | Case=Dat | < https://universaldependencies.org/u/feat/Case#Dat > |
| 6"homini"@la | Gender=Masc | < https://universaldependencies.org/u/feat/Gender#Masc > |
| 7"homini"@la | Number=Sing | < https://universaldependencies.org/u/feat/Number#Sing > |
| 8"homini"@la | POS=NOUN | < https://universaldependencies.org/u/pos/NOUN > |



Conclusion

Thank you



Giovanna Tonazzo - 5216819



giovanna.tonazzo01@icatt.it